

NAG Library Routine Document

F07BRF (ZGBTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07BRF (ZGBTRF) computes the LU factorization of a complex m by n band matrix.

2 Specification

```
SUBROUTINE F07BRF(M, N, KL, KU, AB, LDAB, IPIV, INFO)
INTEGER          M, N, KL, KU, LDAB, IPIV(min(M,N)), INFO
complex*16      AB(LDAB,*)
```

The routine may be called by its LAPACK name *zgbtrf*.

3 Description

F07BRF (ZGBTRF) forms the LU factorization of a complex m by n band matrix A using partial pivoting, with row interchanges. Usually $m = n$, and then, if A has k_l nonzero subdiagonals and k_u nonzero superdiagonals, the factorization has the form $A = PLU$, where P is a permutation matrix, L is a lower triangular matrix with unit diagonal elements and at most k_l nonzero elements in each column, and U is an upper triangular band matrix with $k_l + k_u$ superdiagonals.

Note that L is not a band matrix, but the nonzero elements of L can be stored in the same space as the subdiagonal elements of A . U is a band matrix but with k_l additional superdiagonals compared with A . These additional superdiagonals are created by the row interchanges.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- | | | |
|----|---|--------------|
| 1: | M – INTEGER | <i>Input</i> |
| | <i>On entry:</i> m , the number of rows of the matrix A . | |
| | <i>Constraint:</i> $M \geq 0$. | |
| 2: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of columns of the matrix A . | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 3: | KL – INTEGER | <i>Input</i> |
| | <i>On entry:</i> k_l , the number of subdiagonals within the band of the matrix A . | |
| | <i>Constraint:</i> $KL \geq 0$. | |
| 4: | KU – INTEGER | <i>Input</i> |
| | <i>On entry:</i> k_u , the number of superdiagonals within the band of the matrix A . | |
| | <i>Constraint:</i> $KU \geq 0$. | |

5: AB(LDAB,*) – **complex*16** array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, N)$.

On entry: the m by n coefficient matrix A .

The matrix is stored in rows $k_l + 1$ to $2k_l + k_u + 1$; the first k_l rows need not be set, more precisely, the element A_{ij} must be stored in

$$AB(k_l + k_u + 1 + i - j, j) = A_{ij} \quad \text{for } \max(1, j - k_u) \leq i \leq \min(m, j + k_l).$$

On exit: if $\text{INFO} \geq 0$, AB is overwritten by details of the factorization.

The upper triangular band matrix U , with $k_l + k_u$ superdiagonals, is stored in rows 1 to $k_l + k_u + 1$ of the array, and the multipliers used to form the matrix L are stored in rows $k_l + k_u + 2$ to $2k_l + k_u + 1$.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07BRF (ZGBTRF) is called.

Constraint: $\text{LDAB} \geq 2 \times \text{KL} + \text{KU} + 1$.

7: IPIV($\min(M, N)$) – INTEGER array *Output*

On exit: the pivot indices. Row i of the matrix A was interchanged with row $\text{IPIV}(i)$, for $i = 1, 2, \dots, \min(m, n)$.

8: INFO – INTEGER *Output*

On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If $\text{INFO} = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

If $\text{INFO} = i$, $U(i, i)$ is exactly zero. The factorization has been completed, but the factor U is exactly singular, and division by zero will occur if it is used to solve a system of equations.

7 Accuracy

The computed factors L and U are the exact factors of a perturbed matrix $A + E$, where

$$|E| \leq c(k)\epsilon P|L||U|,$$

$c(k)$ is a modest linear function of $k = k_l + k_u + 1$, and ϵ is the **machine precision**. This assumes $k \ll \min(m, n)$.

8 Further Comments

The total number of real floating-point operations varies between approximately $8nk_l(k_u + 1)$ and $8nk_l(k_l + k_u + 1)$, depending on the interchanges, assuming $m = n \gg k_l$ and $n \gg k_u$.

A call to F07BRF (ZGBTRF) may be followed by calls to the routines:

F07BSF (ZGBTRS) to solve $AX = B$, $A^T X = B$ or $A^H X = B$;

F07BUF (ZGBCON) to estimate the condition number of A .

The real analogue of this routine is F07BDF (DGBTRF).

9 Example

This example computes the LU factorization of the matrix A , where

$$A = \begin{pmatrix} -1.65 + 2.26i & -2.05 - 0.85i & 0.97 - 2.84i & 0.00 + 0.00i \\ 0.00 + 6.30i & -1.48 - 1.75i & -3.99 + 4.01i & 0.59 - 0.48i \\ 0.00 + 0.00i & -0.77 + 2.83i & -1.06 + 1.94i & 3.33 - 1.04i \\ 0.00 + 0.00i & 0.00 + 0.00i & 4.48 - 1.09i & -0.46 - 1.72i \end{pmatrix}.$$

Here A is treated as a band matrix with one subdiagonal and two superdiagonals.

9.1 Program Text

```
*      F07BRF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, KLMAX, KUMAX, LDAB
      PARAMETER        (MMAX=8,NMAX=8,KLMAX=8,KUMAX=8,
+                      LDAB=2*KLMAX+KUMAX+1)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, K, KL, KU, M, N
*      .. Local Arrays ..
      COMPLEX *16      AB(LDAB,NMAX)
      INTEGER          IPIV(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
      EXTERNAL          X04DFF, ZGBTRF
*      .. Intrinsic Functions ..
      INTRINSIC         MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07BRF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N, KL, KU
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. KL.LE.KLMAX .AND. KU.LE.KUMAX)
+      THEN
*
*      Read A from data file
*
      K = KL + KU + 1
      READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,M)
*
*      Factorize A
*
      CALL ZGBTRF(M,N,KL,KU,AB,LDAB,IPIV,INFO)
*
*      Print details of factorization
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DFF(M,N,KL,KL+KU,AB,LDAB,'Bracketed','F7.4',
+          'Details of factorization','Integer',RLABS,
+          'Integer',CLABS,80,0,IFAIL)
*
*      Print pivot indices
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'IPIV'
      WRITE (NOUT,99999) (IPIV(I),I=1,MIN(M,N))
*
      IF (INFO.NE.0) WRITE (NOUT,*) 'The factor U is singular'
```

```

*
      END IF
*
99999 FORMAT ((1X,I12,3I18))
      END

```

9.2 Program Data

F07BRF Example Program Data

```

  4  4  1  2                                     :Values of M, N, KL and KU
(-1.65, 2.26) (-2.05,-0.85) ( 0.97,-2.84)
( 0.00, 6.30) (-1.48,-1.75) (-3.99, 4.01) ( 0.59,-0.48)
              (-0.77, 2.83) (-1.06, 1.94) ( 3.33,-1.04)
                                   ( 4.48,-1.09) (-0.46,-1.72) :End of matrix A

```

9.3 Program Results

F07BRF Example Program Results

Details of factorization

	1	2	3	4
1	(0.0000, 6.3000)	(-1.4800,-1.7500)	(-3.9900, 4.0100)	(0.5900,-0.4800)
2	(0.3587, 0.2619)	(-0.7700, 2.8300)	(-1.0600, 1.9400)	(3.3300,-1.0400)
3		(0.2314, 0.6358)	(4.9303,-3.0086)	(-1.7692,-1.8587)
4			(0.7604, 0.2429)	(0.4338, 0.1233)

IPIV

2

3

3

4
